

# Databound Grid – Data Abstraction Extended

By Jim Douglas

T

he BBJDataBoundGrid provides an easy way to display and edit data from a BASIS multikeyed file (MKEYED, XKEYED or VKEYED) or from any SQL-accessible database. It builds on recordsets and databound controls first introduced in BBJ 4.0. [Databound controls](#) define the association between fields in the database and

GUI controls on the screen. With the databound controls automatically transferring data between the screen and the database, the developer can focus on application details to build data-oriented applications more quickly and easily than ever before.



code	name
NM	New Mexico
NV	Nevada
NY	New York
OH	Ohio
OK	Oklahoma
OR	Oregon

## File Maintenance Sample

**Figure 1** shows a typical grid-based maintenance screen for viewing and editing state codes.

**Figure 2** shows the complete source that produced this maintenance screen. [continued...](#)

Figure 1. BBJDataBoundGrid state table

```
1 rem ' BBJDataBoundGrid State Table
2
3 rem ' (1) Create a BBJRecordSet
4 database$ = "ChileCompany"
5 options$ = "user=admin,password=admin123"
6 select$ = "select * from state"
7 recordset! = BBJAPI().createSQLRecordSet(database$,options$,select$)
8
9 rem ' (2) Create a BBJWindow with a BBJDataBoundGrid
10 sysgui! = bbjapi().openSysGui("X0")
11 window! = sysgui!.addWindow(100,100,200,200,"States", $00110083$)
12 window!.setCallback(window!.ON_CLOSE,"eoj")
13 dbgrid! = window!.addDataBoundGrid(101,10,10,180,180,$81ce$)
14 dbgrid!.setColumnHeaderAlignment(dbgrid!.GRID_ALIGN_LEFT)
15 dbgrid!.setRowHeight(25)
16 dbgrid!.setEditable(1)
17 dbgrid!.setFitToGrid(1)
18 dbgrid!.focus()
19
20 rem ' (3) Bind the BBJRecordSet to the BBJDataBoundGrid
21 dbgrid!.bindRecordSet(recordset!)
22 dbgrid!.setDefaultColumnHeaders()
23 dbgrid!.setDefaultAlignment(dbgrid!.GRID_ALIGN_LEFT)
24
25 rem ' (4) On requested row change, update any edits
26 dbgrid!.setCallback(dbgrid!.ON_DB_GRID_ROW_CHANGE_REQUEST,"Row_Change")
27
28 process_events
29
30 eoj:
31 release
32
33 Row_Change:
34 rem ' (4) Update the changes to the underlying database
35 if (dbgrid!.isEditable() and recordset!.isCurrentRecordDirty()) then
36   recorddata! = recordset!.getCurrentRecordData()
37   recordset!.update(recorddata!,err="next")
38 endif
39 rem ' Move to the requested row and column
40 rowchange! = sysgui!.getLastEvent()
41 recordset!.moveToRecord(rowchange!.getRow())
42 dbgrid!.setSelectedColumn(rowchange!.getColumn())
43 return
```

Figure 2. BBJDataBoundGrid state table – listing (State.src)



Jim Douglas  
Software Engineer  
Contractor


The main sections of this program are:

- (1) **Create a BBJ RecordSet**  
Uses an SQLRecordSet that maps to the state table of the ChileCompany database, which is an MKEYED file with an associated entry in the Data Dictionary. Recordsets can also be associated with BBJ multikeyed files (FileRecordSet) via a string template or with memory-based data structures (MemoryRecordSet) that might have been constructed from READ or READ RECORD statements applied to non-normalized data files.
- (2) **Create a BBJ Window with a BBJ DataBoundGrid**
- (3) **Bind the BBJ RecordSet to the BBJ DataBoundGrid**  
Associates the databound grid to the recordset and creates an ongoing two-way linkage between the grid and the underlying data structure.
- (4) **Update the changes to the underlying database**  
When the user attempts to move to a new row, BBJ fires a BBJDBGridRowChangeEvent for which the programmer may choose how to respond. This sample calls BBJRecordSet::isCurrentRecordDirty() to see if any changes were made; if so, it writes the changed record to the underlying database. Finally, it moves to the requested row and column.

To see the two-way linkage in action, run two copies of State.src at the same time. Change one of the state names on one screen and move away from that row to commit the change to the database. When navigating again to that row on the other screen, the value updates automatically to reflect the change.

### Property Sheet Sample

Databound grids are a convenient way to display any table-oriented data. **Figure 3** displays a typical property sheet that shows current Java system properties. The source code list in **Figure 4** uses a memory-based databound grid for the property sheet sample.



property	value
java.runtime.name	Java(TM) SE Runtime Environment
java.runtime.version	1.6.0_06-b02
java.specification.name	Java Platform API Specification
java.specification.vendor	Sun Microsystems Inc.
java.specification.version	1.6
java.util.logging.config.file	c:/util/logging.properties
java.vendor	Sun Microsystems Inc.
java.vendor.url	http://java.sun.com/

**Figure 3.** BBJDataBoundGrid property sheet

The main sections of this program are similar to the first sample:

- (1) **Create a BBJ RecordSet**  
Uses a MemoryRecordSet of properties and corresponding values.
- (2) **Populate the BBJ RecordSet with the Java system properties**  
Lists the current Java system properties in alphabetical order.
- (3) **Create a BBJ Window with a BBJ DataBoundGrid**
- (4) **Bind the BBJ RecordSet to the BBJ DataBoundGrid**  
Binds (associates) the databound grid to the recordset.
- (5) **Accept any requested row changes**  
Fires a BBJDBGridRowChangeEvent when the user attempts to move to a new row. This sample just displays the data without needing to allow for updates and moves to the requested row and column.

*continued...*

```

1 rem ' BBJDataBoundGrid Property Sheet
2
3 rem ' (1) Create a BBJRecordSet
4 template$ = "property:c(1*),value:c(1*)"
5 recordset! = bbjapi().createMemoryRecordSet(Template$)
6
7 rem ' (2) Populate the BBJRecordSet with the Java system properties
8 p! = java.lang.System.getProperties()
9 k! = new java.util.TreeMap(p!)
10 i! = k!.keySet().iterator()
11 while i!.hasNext()
12     key! = i!.next()
13     val! = p!.get(key!)
14     rec! = recordset!.getEmptyRecordData()
15     rec!.setFieldValue("property",str(key!))
16     rec!.setFieldValue("value",str(val!))
17     recordset!.insert(rec!)
18 wend
19
20 rem ' (3) Create a BBJWindow with a BBJDataBoundGrid
21 sysgui! = bbjapi().openSysGui("X0")
22 window! = sysgui!.addWindow(100,100,400,500,"System Properties",,$00110083$)
23 window!.setCallback(window!.ON_CLOSE,"eoj")
24 window!.setCallback(window!.ON_RESIZE,"resize")
25 dbgrid! = window!.addDataBoundGrid(101,10,10,380,480,$81ce$)
26 dbgrid!.setFont(sysgui!.makeFont("Tahoma",11,1))
27 dbgrid!.setColumnHeaderAlignment(dbgrid!.GRID_ALIGN_LEFT)
28 dbgrid!.setRowHeight(25)
29 dbgrid!.setFitToGrid(1)
30 dbgrid!.focus()
31
32 rem ' (4) Bind the BBJRecordSet to the BBJDataBoundGrid
33 dbgrid!.bindRecordSet(recordset!)
34 dbgrid!.setDefaultColumnHeaders()
35 dbgrid!.setDefaultAlignment(dbgrid!.GRID_ALIGN_LEFT)
36
37 rem ' (5) Accept any requested row changes
38 dbgrid!.setCallback(dbgrid!.ON_DB_GRID_ROW_CHANGE_REQUEST,"Row_Change")
39
40 process_events
41
42 eoj:
43 release
44
45 Row_Change:
46     rem ' (5) Move to the requested row and column
47     rowchange! = sysgui!.getLastEvent()
48     recordset!.moveToRecord(rowchange!.getRow())
49     dbgrid!.setSelectedColumn(rowchange!.getColumn())
50 return
51
52 resize:
53     rem ' Resize the grid when the screen is resized
54     event! = sysgui!.getLastEvent()
55     dbgrid!.setSize(event!.getWidth()-20,event!.getHeight()-20)
56 return

```

Figure 4. BBJDataBoundGrid property sheet – listing (Properties.src)

## Database Table Editor Sample

The final sample program uses a databound grid to implement a powerful and flexible database editor as displayed in **Figure 5** and **Figure 6**. In addition to being able to change individual fields by editing grid cells, the user can also add, change, or delete records using a dynamically created full-screen editor.

## Databound vs. Data-Aware

The databound grid is more powerful and flexible than the older data-aware grid. The data-aware grid automatically writes any changes to the underlying database, without possible programmer intervention. This can allow for slightly less programming, but at the cost of flexibility. With the databound grid, developers always choose when to update changes to the database, allowing more control over data integrity. Furthermore, they can use the databound grid in coordination with other databound controls, all interacting with a single recordset.

*continued...*




Figure 5. Database table editor (DBGri d. src)




Figure 6. Database table editor – full-screen record editor

## Summary

When using databound controls, BBJ automatically manages the housekeeping details of copying data from the database record to the screen, and back again. This leaves developers free to focus on the business rules, reducing the time it takes to build powerful and flexible data-oriented applications.



Download the code samples for this article from [www.basis.com/advantage/mag-v12n1/databoundgrid.zip](http://www.basis.com/advantage/mag-v12n1/databoundgrid.zip)

Refer to the online documentation at [www.basis.com/onlinedocs/documentation/flash](http://www.basis.com/onlinedocs/documentation/flash) and search the index for:

[BBjRecordSet](#)  
[BBjDataBoundGrid](#)

Recordsets provide a consistent interface to the data in a BBJ multikeyed file, SQL database, or (as of BBJ 7.0) an in-memory data structure. For more information about the recordsets, see:

*Using the BBjRecordSet*  
[www.basis.com/advantage/mag-v7n3/bbjrecordset.pdf](http://www.basis.com/advantage/mag-v7n3/bbjrecordset.pdf)

*Why Use the BBjRecordSet?*  
[www.basis.com/advantage/mag-v8n1/recordset.html](http://www.basis.com/advantage/mag-v8n1/recordset.html)

Databound controls are BBJ GUI controls (e.g. ListButton, InputE, InputN) with an ongoing link to a field in a recordset. For more information about databound controls, review:

*BBjDataBound Controls*  
[www.basis.com/advantage/mag-v7n3/bbjdataboundcontrols.pdf](http://www.basis.com/advantage/mag-v7n3/bbjdataboundcontrols.pdf)

*Watch the Form Gen Wizard Trans"form" Data*  
[www.basis.com/advantage/mag-v11n1/FormGenWizard01adv07\\_Links.pdf](http://www.basis.com/advantage/mag-v11n1/FormGenWizard01adv07_Links.pdf)